```
;*********************************************************
;                                                        *
;       PROGRAM ID:     DISK CONTROLLER MODULE           *
;                                                        *
;       VERSION:        2.2  8"      RELEASE 2A          *
;                                                        *
;*********************************************************
;                                                        *
;       PRESENTED BY:   JADE COMPUTER PRODUCTS INC.      *
;                       4901 W. ROSECRANS BLVD.          *
;                       HAWTHORNE, CALIFORNIA            *
;                       90250,   U.S.A.                  *
;                                                        *
;*********************************************************
;                                                        *
;       WRITTEN BY:     STAN KRUMME                      *
;                                                        *
;*********************************************************
; THE DISK CONTROLLER MODULE (DCM2) EXECUTES INTERNAL    *
; TO THE  JADE DOUBLE D DISK CONTROLLER BOARD.  THIS     *
; PROGRAM PROVIDES A FACILITY TO READ/WRITE DISKETTE     *
; SECTORS AND FORMAT DISKETTE TRACKS  (IN SINGLE AND     *
; DOUBLE DENSITY).  THIS DCM SETS THE PARAMETERS FOR     *
; EACH  DRIVE  DURING THE  "LOG-ON" OPERATION.   THE     *
; FORMAT.COM PROGRAM WRITES AN IDENTIFICATION SECTOR     *
; WHICH PROVIDES  THE NEEDED INFORMATION.   IF  THIS     *
; IDENTITY  SECTOR  IS NOT PRESENT  ON THE DISKETTE,     *
; IT IS  ASSUMED TO BE A  STANDARD  8"  3740 FORMAT.     *
; THIS PROGRAM CONTAINS A 4 WORD  TIMING BLOCK WHICH     *
; SHOULD BE PATCHED TO MATCH THE  USERS DISK DRIVES.     *
; THIS HAS NORMALLY BEEN SET FOR  SHUGART SA800/801.     *
;*********************************************************

;*********************************************************
; DISK CONTROLLER MODULE IS  COMMAND  COMPATABLE WITH    *
; THE FOLLOWING  WESTERN  DIGITAL  CONTROLLER  CHIPS.    *
; DOUBLE D  USER SWITCH 0  (U0 OR R0)  MUST BE SET TO    *
; INDICATE THE  CONTROLLER CHIP  DATA  BUS  POLARITY.    *
;*********************************************************
;       CONTROLLER IC           USER SW0                 *
;       -------------           --------                 *
;       FD1791-02 (01)           CLOSED                  *
;       FD1793-02 (01)           OPENED                  *
;       FD1795-02                CLOSED                  *
;       FD1797-02                OPENED                  *
;*********************************************************
; THE FD1795-02 AND FD1797-02 PROVIDE ENHANCED SINGLE    *
; DENSITY PERFORMANCE IN THAT  THESE  CHIPS ARE FULLY    *
; COMPATABLE WITH FD1771-01 3740 FORMATS.                *
;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING IS A LIST OF THE INTERNAL I/O ADDRESS *
                        ; ASSIGNMENTS.   THESE PORTS AND CONTROLS CAN ONLY BE *
                        ; USED BY THE ONBOARD Z80A.  THESE PORTS AND CONTROLS *
                        ; ARE NOT IN THE S100 BUS ADDRESS SPACE.              *
                        ;*********************************************************

                        ;*******( CONTROLLER PORT ASSIGNMENTS )*****************

0000                    BL.STS  ==      000H    ;BOARD STATUS PORT.
0000                    BL.CTL  ==      000H    ;BOARD CONTROL PORT.
0004                    WD.CMD  ==      004H    ;179X COMMAND REGISTER.
0004                    WD.STS  ==      004H    ;179X STATUS REGISTER.
0005                    WD.TRK  ==      005H    ;179X TRACK REGISTER.
0006                    WD.SEC  ==      006H    ;179X SECTOR REGISTOR.
0007                    WD.DTA  ==      007H    ;179X DATA REGISTER.

                        ;*******( CONTROLLER FUNCTION ASSIGNMENTS )*************

0008                    XP.STP  ==      008H    ;ISSUE STEP PULSE.
0010                    XP.MTO  ==      010H    ;MOTOR TURN OFF.
0020                    XP.IRR  ==      020H    ;S100 INT-REQ RESET.
0040                    XP.MTX  ==      040H    ;MOTOR TIME EXTEND.
0080                    XP.DSH  ==      080H    ;DATA SYNC HOLD.

                        ;*********************************************************
                        ; THE FOLLOWING LIST ASSIGNS EACH  BIT  POSITION  AND *
                        ; FUNCTION OF THE BOARD CONTROL PORT (BL.CTL).        *
                        ;*********************************************************

                        ;*******( BIT ASSIGNMENTS )*****************************

0001                    BC.DSA  ==  00000001B   ;DRIVE SELECT A (2*0).
0002                    BC.DSB  ==  00000010B   ;DRIVE SELECT B (2*1).
0004                    BC.DSE  ==  00000100B   ;DRIVE SELECT ENABLE.
0008                    BC.EIA  ==  00001000B   ;EIA SIGNAL LEVEL OUT.
0010                    BC.DDE  ==  00010000B   ;DOUBLE DENSITY ENABLE.
0020                    BC.DAS  ==  00100000B   ;DIRECTION AND SIDE
0040                    BC.PCA  ==  01000000B   ;PRECOMP SELECT A.
0080                    BC.PCB  ==  10000000B   ;PRECOMP SELECT B.

                        ;*******( FUNCTION ASSIGNMENTS )***********************

0003                    BC.DSN  ==  BC.DSA!BC.DSB       ;DRIVE NMBR MASK.
0000                    BC.SDS  ==  0                   ;SINGLE DENSITY.
0010                    BC.DDS  ==  BC.DDE              ;DOUBLE DENSITY.
0040                    BC.PCH  ==  BC.PCA              ;PRECOMP - HEAVY.
0080                    BC.PCM  ==  BC.PCB              ;PRECOMP - MEDIUM.
00C0                    BC.PCL  ==  BC.PCA!BC.PCB       ;PRECOMP - LIGHT.
0020                    BC.SD1  ==  BC.DAS              ;SELECT SIDE ONE.
0020                    BC.INW  ==  BC.DAS              ;STEP INWARD DIRC.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING LIST DEFINES EACH BIT AND FUNCTION OF *
                        ; THE BOARD STATUS PORT (BL.STS).                      *
                        ;*********************************************************

0001                    BS.US0  ==  00000001B   ;USER SWITCH 0.
0002                    BS.US1  ==  00000010B   ;USER SWITCH 1.
0004                    BS.TST  ==  00000100B   ;TEST MODE SWITCH.
0008                    BS.INT  ==  00001000B   ;HOST INT REQUEST.
0010                    BS.EIA  ==  00010000B   ;EIA SIGNAL LEVEL IN.
0020                    BS.MOF  ==  00100000B   ;MOTOR OFF INIDCATOR.
0040                    BS.TSD  ==  01000000B   ;TWO SIDED DRIVE FLAG.
0080                    BS.DCN  ==  10000000B   ;DISK CHANGE INDICATOR.

                        ;*********************************************************
                        ; THE FOLLOWING IS A LIST OF  COMMAND CODES ISSUED TO *
                        ; THE 179X-02 DISK CONTROLLER.                         *
                        ;*********************************************************

0018                    DC.HDL  ==  00011000B   ;SEEK/LOAD RW HEAD.
0010                    DC.HDU  ==  00010000B   ;SEEK/UNLD RW HEAD.
0088                    DC.RDS  ==  10001000B   ;READ SECTOR.
00A8                    DC.WRS  ==  10101000B   ;WRITE SECTOR.
00F0                    DC.WRT  ==  11110000B   ;WRITE TRACK FORMAT.
00C0                    DC.RDA  ==  11000000B   ;READ TRACK ADDRESS.
00D0                    DC.STS  ==  11010000B   ;SET TYPE 1 STATUS
00D8                    DC.IFI  ==  11011000B   ;FORCED INTERRUPT.

                        ;*********************************************************
                        ; THE FOLLOWING LIST  CONTAINS ALL THE  MASKS USED TO *
                        ; TEST THE 179X-02 STATUS CODES  (PORT WD.STS).        *
                        ;*********************************************************

009D                    DM.RER  ==  10011101B   ;READ ERROR TEST.
00FD                    DM.WER  ==  11111101B   ;WRITE ERROR TEST.
00E4                    DM.FER  ==  11100100B   ;FORMAT ERROR TEST.
0004                    DM.TK0  ==  00000100B   ;TRACK 0 TEST.
0020                    DM.HDL  ==  00100000B   ;HEAD LOAD TEST.
0080                    DM.DNR  ==  10000000B   ;DRIVE NOT READY.
0004                    DM.LDE  ==  00000100B   ;LOST DATA ERROR.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING LIST DEFINES INTERNAL MEMORY.          *
                        ;*********************************************************

                        ;***********( BASE ADDRESS FOR DCM )*********************

1000                    BASE    ==      1000H           ;BASE ADDRESS.

                        ;***********( MEMORY BANKS )*****************************

1000                    BANK.0  ==  BASE+0000H          ;BANK 0 DEFINED.
0400                    BANK.L  ==  0400H               ;BANK LENGTH.
1400                    BANK.1  ==  BANK.0+BANK.L       ;BANK 1 DEFINED.

                        ;************( RESTART VECTORS )************************

1000                    RST.0   ==   BANK.0+0000H       ;RESTART 0.
1008                    RST.1   ==   BANK.0+0008H       ;RESTART 1.
1010                    RST.2   ==   BANK.0+0010H       ;RESTART 2.
1018                    RST.3   ==   BANK.0+0018H       ;RESTART 3.
1020                    RST.4   ==   BANK.0+0020H       ;RESTART 4.
1028                    RST.5   ==   BANK.0+0028H       ;RESTART 5.
1030                    RST.6   ==   BANK.0+0030H       ;RESTART 6.
1038                    RST.7   ==   BANK.0+0038H       ;RESTART 7.

                        ;***********( INTERRUPT VECTORS )************************

1038                    HR.INT  ==  RST.7               ;MASKABLE.
1066                    NM.INT  ==  BANK.0+0066H        ;NON MASKABLE.

                        ;***********( I/O COMMUNICATION  )************************

1370                    IO.BLK  ==  BANK.0+0370H        ;I/O BLOCK BEGIN.
1370                    TP.STK  ==  IO.BLK+0000H        ;TOP OF STACK.
1370                    CMD.BK  ==  IO.BLK+0000H        ;COMMAND BLOCK.
1380                    BUF.BG  ==  IO.BLK+0010H        ;SECTOR BUFFER.
1700                    FMT.BG  ==  BANK.1+0300H        ;FORMAT BUFFER.
1708                    FMT.PS  ==  FMT.BG+0008H        ;FORMAT PROGRAM.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; WAIT IS A "RESTART" TO THE TIMER SUBROUTINE ENTRY.    *
                        ; THIS SUBROUTINE PROVIDES MOST OF THE TIMING USED BY  *
                        ; THE DOUBLE D CONTROLLER.                             *
                        ;*********************************************************

                                .DEFINE WAIT = [
                                 RST    1]

                        ;*********************************************************
                        ; ASSEMBLER DIRECTIVES                                  *
                        ;*********************************************************

                                .PABS           ;ABSOLUTE ADDRESSING.
                                .PHEX           ;INTEL HEX OBJECT FILE.
                                .XLINK          ;NO LINKAGE REQUIRED.

                        ;*********************************************************
                        ; TENTH MILLESECOND TIMING CONSTANTS / 0.2 MS FOR 5"    *
                        ;*********************************************************

0019                    TMR.FC  ==      0019H   ;TIMING CONSTANT, FIRST PASS.
001C                    TMR.NC  ==      001CH   ;TIMING CONSTANT, REPEAT PASS.

                        ;*********************************************************
                        ;       BAUD RATE GENERATOR - TIMING CONSTANTS          *
                        ;*********************************************************
                        ;   BAUDRATE      US/BIT       8" SYS     5 " SYS      *
                        ;   --------      ------       ------     -------      *
                        ;    19200         52.1          9         N.A.        *
                        ;     9600        104.2         25          9          *
                        ;     4800        208.3         57         25          *
                        ;     2400        416.6        121         57          *
                        ;     1200        833.3        248        121          *
                        ;      600       1666.6        N.A.       248          *
                        ;*********************************************************

0019                    BAUD.C  ==      25.     ;BAUD RATE CONSTANT 9600 8".

                        ;*********************************************************
                        ; ERROR RECOVERY VALUES                                 *
                        ;*********************************************************

0005                    RTY.SK  ==      5       ;REPOSITION R/W HEAD ON RETRY.
0009                    RTY.LS  ==      9       ;LAST REPEATED RETRY.

001A                    TRK.OB  ==      26      ;AT FIRST THIRD TRACK OF DISK.
0034                    TRK.IB  ==      52      ;AT SECOND THIRD TRACK.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING AREA IS THE INITIAL START JUMP TABLE.  *
                        ; THE FIRST JUMP IS EXECUTED WHEN THE ONBOARD Z80A IS  *
                        ; RESET.   THE  SECOND JUMP IS THE  DCM ENTRY  FROM A  *
                        ; BOOTSTRAP LOADER.  THIS ENTRY ASSUMES  DCM HAS BEEN  *
                        ; LOADED INTO DOUBLE D BANK 1  BY THE LOADER ROUTINE.  *
                        ; THE LAST  TWO BYTES  HOLD THE  JUMP ADDRESS USED BY  *
                        ; RESTART INTERRUPT ROUTINE AT BANK 0 + 0380H.         *
                        ;*********************************************************

1000                            .LOC    RST.0           ;MODULE BEGINNING.

1000    C3 0000                  JMP    0               ;NOT IMPLEMENTED.
1003    C3 1780                  JMP    INIT.B+BANK.L   ;BOOTSTRAPPED ENTRY.
1006    1041            HR.VEC: .WORD   X.CUTE          ;HOST INTERRUPT VECTOR.

                        ;*********************************************************
                        ; THE FOLLOWING SUBROUTINE IS THE ENTRY POINT FOR THE  *
                        ; DISK CONTROLLER TIMING MODULE. THIS MODULE PROVIDES  *
                        ; DELAYS WHICH ARE MULTIPLES OF 100 MICROSECONDS. THE  *
                        ; CONTENTS OF REGISTER PAIR  DE  DETERMINES THE TOTAL  *
                        ; PERIOD.   (DELAY = (DE )* 100 MICROSECONDS).   THIS  *
                        ; SUBROUTINE IS ENTERED BY THE MACRO "WAIT".           *
                        ;*********************************************************

1008                            .LOC    RST.1           ;TIMING ENTRY POINT.

1008    0619                    MVI     B,TMR.FC        ;FIRST TICK CONSTANT.
100A    10FE                    DJNZ    .               ;AUTO DEC UNTIL ZERO.
100C    C3 1074                 JMP     TICK.E          ;JUMP TO TICK ENTRY.

                        ;*********************************************************
                        ; THE FOLLOWING SECTION IS THE DISK DRIVE TIMING AREA. *
                        ; THE TIMES ARE SET FOR THE SHUGART SA800.  THIS AREA  *
                        ; SHOULD BE MODIFIED FOR THE END USERS DRIVE TYPE.     *
                        ;*********************************************************

                        ;*******( TIMING VALUES IN 0.1 MS )*********************

1010                            .LOC    RST.2

1010    015E            TM.HLD: .WORD   350     ;HEAD ENGAGE TIME.
1012    0050            TM.STP: .WORD   80      ;STEPPER INTERVAL.
1014    0050            TM.ALS: .WORD   80      ;AFTER LAST STEP.
1016    0001            TM.MTO: .WORD   1       ;MOTOR START UP.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING SUBROUTINE PROVIDES THE R/W HEAD CNTL    *
                        ; FUNCTION.  AS THE  FD179X-02  DOES  NOT  OFFER THIS   *
                        ; EXPLICIT COMMAND, THE SEEK COMMAND (TYPE-1) IS USED   *
                        ; WITH THE HEAD LOAD BIT SET / RESET. THE DESTINATION   *
                        ; TRACK IS SET EQUAL TO THE TRACK REGISTER  TO BYPASS   *
                        ; THE  FD179X-02 STEPPING FUNCTION.   PLEASE REFER TO   *
                        ; THE FD179X-02 FLOW-CHART FOR TYPE-1 COMMANDS.         *
                        ;*********************************************************

1018    FDE1            EX.HCF: POP     Y               ;RETURN ADDR IN REG Y.
101A    DB05                    IN      WD.TRK          ;READ PRESENT TRACK.
101C    D307                    OUT     WD.DTA          ;SET DESTINATION TRK.
101E    78                      MOV     A,B             ;LOAD TYPE-1 COMMAND.
101F    A9                      XRA     C               ;INVERT (1791-01).
1020    D304                    OUT     WD.CMD          ;ISSUE COMMAND.
1022    18FE                    JMPR    .               ;WAIT FOR INTERRUPT.

                        ;*********************************************************
                        ; THE  FOLLOWING  SUBROUTINE  UPDATES  THE  FD179X-02   *
                        ; STATUS PORT TO REFLECT CURRENT TYPE-1 STATUS CODES.    *
                        ; NOTE: THIS IS A  TYPE-4  COMMAND WITH  NO INTERRUPT   *
                        ; CONDITIONS SET.                                       *
                        ;*********************************************************

1024    3ED0            EX.STS: MVI     A,DC.STS        ;LOAD SET-STATUS CMND.
1026    A9                      XRA     C               ;INVERT (1791-01).
1027    D304                    OUT     WD.CMD          ;ISSUE COMMAND.
1029    E3                      XTHL                    ;PAUSE FOR FD179X-02.
102A    E3                      XTHL                    ;PAUSE MORE.
102B    E3                      XTHL                    ;PAUSE STILL MORE.
102C    E3                      XTHL                    ;PAUSE LAST TIME.
102D    DB04                    IN      WD.STS          ;INPUT STATUS PORT.
102F    A9                      XRA     C               ;INVERT (1791-01).
1030    C9                      RET                     ;RETURN TO USER.

                        ;*********************************************************
                        ; THE  FOLLOWING SECTION IS THE  MASKABLE  INTERRUPT     *
                        ; ROUTINE.  THIS ROUTINE IS EXECUTED WHEN RESTARTING    *
                        ; THE Z80 FROM A HALT.   THE FUNCTIONS ARE RESET THE    *
                        ; DOUBLE D INT REQ FLIP-FLOP,  PUT  THE  INTERRUPTED    *
                        ; ADDR IN REG DE, AND JUMP ADDRESS AT "HR.VEC".         *
                        ;*********************************************************

1038                            .LOC    HR.INT          ;HOST INTERRUPT ADDR.
1038    DB20                    IN      XP.IRR          ;RESET INTERRUPT REQ FF
103A    D1                      POP     D               ;PURGE INTERRUPTED ADDR
103B    2A 1006                 LHLD    HR.VEC          ;LOAD RETURN ADDRESS
103E    E9                      PCHL                    ;JUMP RETURN ADDRESS

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING SECTION HALTS EXECUTION OF THE           *
                        ; ONBOARD Z80A PROCESSOR.  DURING THIS TIME THE HOST    *
                        ; SYSTEM CAN SWITCH THE CONTROLLER MEMORY INTO THE      *
                        ; S100 BUS FOR STATUS CHECK, SETTING COMMAND BLOCK,     *
                        ; AND SECTOR DATA TRANSFERS.                            *
                        ;*********************************************************

103F    FB              FETCH:  EI              ;ENABLE INTERRUPT START
1040    76                      HLT             ;HALT ON-BOARD PROCESSOR

                        ;*********************************************************
                        ; THE FOLLOWING SECTION GAINS CONTROL AFTER THE DISK     *
                        ; CONTROLLER IS INTERRUPTED FROM THE HALT CONDITION.    *
                        ; THIS SECTION BRACNCHES TO THE INDIVIDUAL COMMAND      *
                        ; ROUTINES.  THE COMMAND TABLE CONTAINS THE ADDRESSES   *
                        ; FOR THIS DISTRIBUTION.                                *
                        ;*********************************************************

1041    3A 1370         X.CUTE: LDA     CB.CMD          ;LOAD HOST COMMAND.
1044    E607                    ANI     CM.MSK          ;MASK ANY OPTIONS.
1046    87                      ADD     A               ;GET 2*A VALUE.
1047    1600                    MVI     D,0             ;ZERO D REGISTER.
1049    5F                      MOV     E,A             ;DE NOW TABLE OFFSET.
104A    21 1053                 LXI     H,CM.DTA        ;LOAD TABLE ADDRESS.
104D    19                      DAD     D               ;NOW POINTS TO ENTRY.
104E    5E                      MOV     E,M             ;LOW ORDER ADDR LOAD.
104F    23                      INX     H               ;POINT TO NEXT BYTE.
1050    56                      MOV     D,M             ;HI ORDER ADDRESS.
1051    EB                      XCHG                    ;BRANCH ADDR IN HL.
1052    E9                      PCHL                    ;BRANCH TO COMMAND.

                        ;*********************************************************
                        ; THE FOLLOWING AREA IS  THE  COMMAND  DRIVER  TABLE.    *
                        ; EACH ENTRY POINTS TO THE COMMAND DRIVER ROUTINE.      *
                        ;*********************************************************

1053                    CM.DTA  ==      .               ;COMMAND TABLE.

1053    10AC            ..CM0A: .WORD   $.LGON          ;LOG-ON DRIVE.
1055    107C            ..CM1A: .WORD   $.READ          ;READ SECTOR.
1057    108A            ..CM2A: .WORD   $.WRIT          ;WRITE SECTOR.
1059    1098            ..CM3A: .WORD   $.FORM          ;FORMAT TRACK.
105B    10C7            ..CM4A: .WORD   $.ADDR          ;READ ADDRESS.
105D    10CF            ..CM5A: .WORD   $.LIST          ;LIST OUTPUT.
105F    10D5            ..CM6A: .WORD   $.LSTT          ;LIST STATUS.
1061    10E4            ..CM7A: .WORD   $.IDLE          ;BACKGROUND.

0007                    CM.MSK  ==      007H            ;COMMAND MASK.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING SECTION IS THE NON-MASKABLE INTERRUPT *
                        ; ROUTINE.  UPON 179X-02 COMMAND TERMINATION THE Z80  *
                        ; RECIEVES A NON-MASKABLE INTERRUPT.  THE STATUS PORT *
                        ; IS INTERROGATED AND SAVED (SV.STS).  REGISTER IY    *
                        ; CONTAINS THE RETURN ADDRESS.                        *
                        ;*********************************************************

1066                            .LOC    NM.INT          ;NON-MASKABLE INT.

1066    DB04            WD.INT: IN      WD.STS          ;GET 179X STATUS.
1068    A9                      XRA     C               ;INVERT (1791).
1069    32 1334                 STA     SV.STS          ;SAVE STATUS.
106C    FDE3                    XTIY                    ;EXCHANGE (SP)<>IY.
106E    ED45                    RETN                    ;RETURN AT OLD IY.

                        ;*********************************************************
                        ; THIS SECTION IS THE REMAINDER OF THE TIMING         *
                        ; SECTION ENTERED BY A RESTART 1.  SEE THAT SECTION   *
                        ; FOR THE DESCRIPTION.                                *
                        ;*********************************************************

1070    061C            TICK.R: MVI     B,TMR.NC        ;NORMAL TICK CONSTANT.
1072    10FE                    DJNZ    .               ;AUTO DEC UNTIL ZERO.
1074    1B              TICK.E: DCX     D               ;DECREMENT AMOUNT.
1075    7A                      MOV     A,D             ;GET HIGH ORDER.
1076    B3                      ORA     E               ;AND LOW ORDER.
1077    00                      NOP                     ;TIMING ADJUST.
1078    00                      NOP                     ;TIMING ADJUST.
1079    20F5                    JRNZ    TICK.R          ;REPEAT UNTIL ZERO.
107B    C9                      RET                     ;RETURN TO USER.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ;  $.READ IS THE READ-SECTOR COMMAND CONTROLLER.       *
                        ;*********************************************************

107C    CD 10EF     $.READ: CALL    SELECT          ;SELECT DRIVE ROUTINE.
107F    CD 1131             CALL    SEEK            ;SEEK TRACK, SET CTLS.
1082    2003                JRNZ    ..EXIT          ;DRIVE OR SEEK ERROR.
1084    CD 1204             CALL    RD.SEC          ;READ DISK SECTOR.
1087    C3 103F     ..EXIT: JMP     FETCH           ;GET NEXT COMMAND.

                        ;*********************************************************
                        ;  $.WRIT IS THE WRITE-SECTOR COMMAND CONTROLLER.      *
                        ;*********************************************************

108A    CD 10EF     $.WRIT: CALL    SELECT          ;SELECT DRIVE ROUTINE.
108D    CD 1131             CALL    SEEK            ;SEEK TRACK, SET CTLS.
1090    2003                JRNZ    ..EXIT          ;DRIVE OR SEEK ERROR.
1092    CD 122F             CALL    WR.SEC          ;WRITE DISK SECTOR.
1095    C3 103F     ..EXIT: JMP     FETCH           ;GET NEXT COMMAND.

                        ;*********************************************************
                        ;  $.FORM IS THE FORMAT-TRACK COMMAND CONTROLLER.      *
                        ;*********************************************************

1098    CD 10EF     $.FORM: CALL    SELECT          ;SELECT DRIVE NUMBER.
109B    3A 1373             LDA     CB.SEC          ;LOAD FORMAT FLAGS.
109E    DD7702              MOV     DV.FLG(X),A     ;RESET DRIVE FLAGS.
10A1    CD 1131             CALL    SEEK            ;SEEK TRACK, SET CTLS.
10A4    2003                JRNZ    ..EXIT          ;DRIVE OR SEEK ERROR.
10A6    CD 125A             CALL    WR.TRK          ;WRITE DISK TRACK.
10A9    C3 103F     ..EXIT: JMP     FETCH           ;GET NEXT COMMAND.

                        ;*********************************************************
                        ;  $.LGON IS THE DRIVE LOG-ON COMMAND CONTROLLER       *
                        ;*********************************************************

10AC    CD 10EF     $.LGON: CALL    SELECT          ;SELECT DRIVE NUMBER.
10AF    AF                  XRA     A               ;ZERO REGISTER A.
10B0    32 1372             STA     CB.TRK          ;SET TRACK AT 0.
10B3    3C                  INR     A               ;NOW A REG IS 1.
10B4    32 1373             STA     CB.SEC          ;SET SECTOR TO ID.
10B7    CD 1131             CALL    SEEK            ;SEEK TRACK, SET CTLS.
10BA    2008                JRNZ    ..EXIT          ;DRIVE OR SEEK ERROR.
10BC    CD 1204             CALL    RD.SEC          ;READ ID SECTOR.
10BF    2003                JRNZ    ..EXIT          ;READ ERROR DETECTED.
10C1    CD 12D7             CALL    LOG.ON          ;LOG ON DISK DRIVE.
10C4    C3 103F     ..EXIT: JMP     FETCH           ;GET NEXT COMMAND.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ;  $.ADDR IS THE READ-ADDRESS COMMAND CONTROLLER.      *
                        ;*********************************************************

10C7    3EFF            $.ADDR: MVI     A,0FFH          ;LOAD ALL ONES.
10C9    32 1377                 STA     CB.STS          ;STORE ERRORS.
10CC    C3 103F                 JMP     FETCH           ;NOT IMPLEMENTED.

                        ;*********************************************************
                        ;  $.LIST IS A LIST DEVICE COMMAND CONTROLLER.         *
                        ;*********************************************************

10CF    CD 12F4         $.LIST: CALL    LST.OT          ;SEND CHAR TO LIST.
10D2    C3 103F                 JMP     FETCH           ;GET NEXT COMMAND.

                        ;*********************************************************
                        ;  $.LSTT CHECKS LIST DEVICE STATUS                    *
                        ;*********************************************************

10D5    DB00            $.LSTT: IN      BL.STS          ;GET BOARD STATUS.
10D7    E610                    ANI     BS.EIA          ;TEST READY BIT.
10D9    CA 10DE                 JZ      ..EXIT          ;IF ZERO GOTO EXIT.
10DC    3EFF                    MVI     A,0FFH          ;LOAD ALL ONES.
10DE    32 1377         ..EXIT: STA     CB.STS          ;STORE STATUS.
10E1    C3 103F                 JMP     FETCH           ;GET NEXT COMMAND.

                        ;*********************************************************
                        ;  $.IDLE IS THE IDLE COMMAND CONTROLLER.              *
                        ;*********************************************************

10E4    DB00            $.IDLE: IN      BL.STS          ;INPUT BOARD STATUS.
10E6    E608                    ANI     BS.INT          ;CHECK HOST INTERRUPT.
10E8    28FA                    JRZ     $.IDLE          ;REPEAT IDLE CHECK.
10EA    DB20                    IN      XP.IRR          ;RESET INTERRUPT REQ.
10EC    C3 103F                 JMP     FETCH           ;GET NEXT COMMAND.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING  SUBROUTINE  SELECTS  REQUESTED DRIVE *
                        ; NUMBER 0-3 (A-D). BEFORE DRIVE SELECTION, THE DRIVE *
                        ; MOTOR CONTROL STATE IS TESTED AND ENABLED IF NEEDED.*
                        ; INDEX REGISTER X  IS SET  POINTING TO THE REQUESTED *
                        ; DRIVE TABLE ENTRY.  THE DRIVE IS THEN SELECTED.     *
                        ;*********************************************************

                        ;*******( MOTOR CHECK ROUTINE )*************************

10EF    DB00            SELECT: IN      BL.STS          ;BOARD LEVEL STATUS.
10F1    E620                    ANI     BS.MOF          ;CHECK MOTOR STATE.
10F3    DB40                    IN      XP.MTX          ;START OR EXTEND TIMER.
10F5    2805                    JRZ     ..CKDV          ;IF WAS ON, NO STARTUP.
10F7    ED5B 1016               LDED    TM.MTO          ;MOTOR STARTUP DELAY.
10FB    CF                      WAIT                    ;PROGRAMMABLE DELAY.

                        ;*******( NEW SELECTION CHECK )************************

10FC    3A 1371         ..CKDV: LDA     CB.DRV          ;LOAD DRIVE NUMBER.
10FF    E603                    ANI     BC.DSN          ;GET DRIVE NUMBER.
1101    DDBE00                  CMP     DV.NBR(X)       ;CURRENTLY SELECTED?
1104    C8                      RZ                      ;RETURN IF DRV SAME.

                        ;*******( SET TABLE POINTER )**************************

1105    DD21 1342               LXI     X,DV.TBL        ;DRIVE TABLE ADDR.
1109    11 0004                 LXI     D,DV.DES        ;DRIVE ENTRY SIZE.
110C    3D              ..NEXT: DCR     A               ;DECREMENT DRV NO.
110D    FA 1114                 JM      ..DSLT          ;IF S=1 EXIT.
1110    DD19                    DADX    D               ;POINT NEXT DRIVE.
1112    18F8                    JMPR    ..NEXT          ;TRY THIS DRIVE.

                        ;*******( DESELECT OLD DRIVE )*************************

1114    0610            ..DSLT: MVI     B,DC.HDU        ;LOAD UNLOAD R/W HEAD.
1116    CD 1018                 CALL    EX.HCF          ;FD179X-02 TYPE 1 CMND.
1119    3A 1333                 LDA     SV.CTL          ;BL.CTL LAST ISSUED.
111C    E6FB                    ANI     #BC.DSE         ;DRIVE SELECT DSBLD.
111E    D300                    OUT     BL.CTL          ;ISSUE DESELECT.

                        ;*******( SELECT NEW DRIVE )***************************

1120    E6FC                    ANI     #BC.DSN         ;STRIP OFF DRIVE NMBR.
1122    DDB600                  ORA     DV.NBR(X)       ;OR IN NEW DRIVE NMBR.
1125    D300                    OUT     BL.CTL          ;OUTPUT DRIVE NMBR.
1127    F604                    ORI     BC.DSE          ;SET DRV ENABLE BIT.
1129    D300                    OUT     BL.CTL          ;ENABLE NEW DRIVE.
112B    E607                    ANI     BC.DSN!BC.DSE   ;NOW JUST DRIVE ENBLED.
112D    32 1331                 STA     SV.DRV          ;SAVE DRIVE SELECT.
1130    C9                      RET                     ;DRIVE IS SELECTED.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING SUBROUTINE PERFORMS THE TRACK SEEK     *
                        ; OPERATION.  AFTER THE SEEK OPERATION, THE DENSITY   *
                        ; AND PRE-COMPENSATION CONTROLS ARE SET.              *
                        ;*********************************************************

                        ;*******( HEAD LOADING )*********************************

1131    CD 1024         SEEK:   CALL    EX.STS          ;GET DRIVE STATUS.
1134    E6A0                    ANI     DM.HDL!DM.DNR   ;CHECK HEAD AND READY.
1136    FA 1174                 JM      ..NRDY          ;DRIVE NOT READY EXIT.
1139    C2 1146                 JNZ     ..DTAS          ;BYPASS IF HEAD LOADED.
113C    0618                    MVI     B,DC.HDL        ;HEAD-LOAD COMMAND.
113E    CD 1018                 CALL    EX.HCF          ;EXEC FD179X-02 TYPE 1.
1141    ED5B 1010               LDED    TM.HLD          ;SET HEAD-LOAD DELAY.
1145    CF                      WAIT                    ;PROGRAMMABLE DELAY.

                        ;*******( DETERMINE TRACK NMBR AND SIDE )***************

1146    DB00            ..DTAS: IN      BL.STS          ;INPUT BOARD STATUS.
1148    E640                    ANI     BS.TSD          ;TEST DISK SIDES FLAG.
114A    3A 1372                 LDA     CB.TRK          ;GET LOGICAL TRACK NO.
114D    6F                      MOV     L,A             ;SAVE LOGICAL TRACK.
114E    C2 1152                 JNZ     ..NDBL          ;SKIP IF NOT DBL SIDED.
1151    1F                      RAR                     ;DIV BY 2 DOUBLE SIDE.
1152    32 1335         ..NDBL: STA     PH.TRK          ;STORE PHYSICAL TRACK.
1155    67                      MOV     H,A             ;SAVE PHYSICAL NUMBER.
1156    3A 1331                 LDA     SV.DRV          ;LOAD DRV NMBR ENABLED.
1159    3002                    JRNC    ..SIDO          ;SKIP NEXT IF SIDE 0.
115B    F620                    ORI     BC.SD1          ;OR IN SELECT SIDE 1.
115D    32 1332         ..SIDO: STA     SV.DAS          ;STORE DRV AND SIDE EN.
1160    57                      MOV     D,A             ;SAVE DRV AND SIDE EN.
1161    7C                      MOV     A,H             ;LOAD PHYSICAL NUMBER.
1162    DD9601                  SUB     DV.TRK(X)       ;TRACK OFFSET TESTED.
1165    2021                    JRNZ    ..SEEK          ;IF OFFTRACK, DO SEEK.
1167    DB00                    IN      BL.STS          ;INPUT BOARD STATUS.
1169    E640                    ANI     BS.TSD          ;TEST DISK SIDES FLAG
116B    CA 117B                 JZ     ..DSID           ;GOTO DOUBLE SIDE CTL.

                        ;*******( SINGLE SIDED DISKETTE )***********************

116E    DD7E03          ..SSID: MOV     A,DV.CTL(X)     ;GET PREVIOUS CONTROLS.
1171    C3 11EA                 JMP     ..EXIT          ;SET CONTROLS / EXIT.

                        ;*******( DRIVE NOT READY EXIT )************************

1174    3E80            ..NRDY: MVI     A,CS.DNR        ;DRIVE NOT READY FLAG.
1176    32 1377                 STA     CB.STS          ;STORE ERROR STATUS.
1179    A7                      ANA     A               ;SET NOT ZERO FLAG.
117A    C9                      RET                     ;ERROR EXIT.

                        ;*********************************************************
```

```
                        ;*******( DISKETTE IS DOUBLE SIDED )********************

117B    7C              ..DSID: MOV     A,H             ;GET PHYSICAL TRK NMBR.
117C    A7                      ANA     A               ;TEST IF TRACK ZERO.
117D    283B                    JRZ     ..DCTL          ;IF ZERO, RESET CNTLS.
117F    DD7E03                  MOV     A,DV.CTL(X)     ;LOAD OLD DRV CTLS.
1182    E6DF                    ANI     #BC.SD1         ;STRIP OFF SIDE CMND.
1184    B2                      ORA     D               ;OR IN NEW SIDE CMND.
1185    C3 11EA                 JMP     ..EXIT          ;SET CONTROLS / EXIT.

                        ;*******( SET DIRECTION AND COUNT STEPS )***************

1188    F5              ..SEEK: PUSH    PSW             ;SAVE REG A AND FLGS.
1189    ED5B 1338               LDED    TM.SAW          ;STEP AFTER WRITE.
118D    CF                      WAIT                    ;PROGRAMMABLE DELAY.
118E    F1                      POP     PSW             ;RESTORE A AND FLGS.
118F    380A                    JRC     ..SOUT          ;IF CARRY STEP OUT.
1191    6F              ..SIN:  MOV     L,A             ;MOVE OFFSET TO L.
1192    3A 1331                 LDA     SV.DRV          ;DRIVE SELECT BITS.
1195    F620                    ORI     BC.INW          ;SET STEP DIRC IN.
1197    D300                    OUT     BL.CTL          ;OUTPUT CONTROL.
1199    180B                    JMPR    ..STEP          ;GOTO STEP ROUTINE.
119B    ED44            ..SOUT: NEG                     ;COMPLEMENT OFFSET.
119D    FA 11FD                 JM      ..HOME          ;BETTER HOME DRV.
11A0    6F                      MOV     L,A             ;MOVE OFFSET TO L.
11A1    3A 1331                 LDA     SV.DRV          ;DRIVE SELECT BITS.
11A4    D300                    OUT     BL.CTL          ;SET DIRECTION OUT.
11A6    DB08            ..STEP: IN      XP.STP          ;ISSUE STEP PULSE.
11A8    ED5B 1012               LDED    TM.STP          ;STEP DELAY TIME.
11AC    CF                      WAIT                    ;PROGRAMMABLE DELAY.
11AD    2D                      DCR     L               ;DECREMENT STEPS.
11AE    20F6                    JRNZ    ..STEP          ;REPEAT OPERATION.
11B0    3A 1332                 LDA     SV.DAS          ;LOAD DRV AND SIDE.
11B3    D300                    OUT     BL.CTL          ;OUTPUT CONTROL.
11B5    ED5B 1014               LDED    TM.ALS          ;MORE AFTER LAST STP.
11B9    CF                      WAIT                    ;PROGRAMMABLE DELAY.

                        ;*************************************************************
```

```
                        ;*******( CONTROL DETERMINATION )**********************

11BA    3A 1372     ..DCTL: LDA     CB.TRK          ;LOAD LOGICAL TRACK.
11BD    FE01                CPI     1               ;COMPARE AGAINST 1.
11BF    3820                JRC     ..SDEN          ;TRACK 0 IS SDENS.
11C1    3E04                MVI     A,DF.DTD        ;DATA TRK DENS FLG.
11C3    C2 11C8             JNZ     ..DTST          ;GOTO TEST DENSITY.
11C6    3E02                MVI     A,DF.T1D        ;TRACK 1 DENS FLAG.
11C8    DDA602      ..DTST: ANA     DV.FLG(X)       ;TEST DENSITY FLAGS.
11CB    CA 11E1             JZ      ..SDEN          ;IF ZERO, THEN SDENS.
11CE    3A 1335     ..DDEN: LDA     PH.TRK          ;LOAD PHYSICAL TRACK.
11D1    FE1A                CPI     TRK.OB          ;TEST OUTSIDE BOUNDRY.
11D3    06D0                MVI     B,BC.DDS!BC.PCL ;DDENS AND LOW PRECOMP.
11D5    380C                JRC     ..CTLS          ;SET FOR OUTSIDE TRKS.
11D7    FE34                CPI     TRK.IB          ;TEST INSIDE BOUNDRY.
11D9    0690                MVI     B,BC.DDS!BC.PCM ;DDENS AND MED PRECOMP.
11DB    3806                JRC     ..CTLS          ;JUMP TO CONTROLS SET.
11DD    0650                MVI     B,BC.DDS!BC.PCH ;DDENS AND MAX PRECOMP.
11DF    1802                JMPR    ..CTLS          ;JUMP TO CONTROLS SET.

11E1    06C0        ..SDEN: MVI     B,BC.SDS!BC.PCL ;SDEN AND PC-LOW.

                        ;*******( SET CONTROL VALUES AND EXIT )*****************

11E3    3A 1332     ..CTLS: LDA     SV.DAS          ;GET DRIVE AND SIDE.
11E6    B0                  ORA     B               ;SET PRECOMP AND DENS.
11E7    DD7703              MOV     DV.CTL(X),A     ;SAVE CONTROLS FOR DRV.
11EA    D300        ..EXIT: OUT     BL.CTL          ;OUTPUT CONTROLS.
11EC    32 1333             STA     SV.CTL          ;SAVE THESE CONTROLS.
11EF    3A 1335             LDA     PH.TRK          ;PHYSICAL TRACK NMBR.
11F2    DD7701              MOV     DV.TRK(X),A     ;SET DRIVE TABLE.
11F5    3A 1372             LDA     CB.TRK          ;LOGICAL TRACK NMBR.
11F8    A9                  XRA     C               ;INVERT (1791-01).
11F9    D305                OUT     WD.TRK          ;SET TRACK REGISTER.
11FB    AF                  XRA     A               ;SET ZERO FLAG.
11FC    C9                  RET                     ;RETURN TO CALLER.

                        ;*******( CALIBRATE TRACK NUMBER )**********************

11FD    CD 12A6     ..HOME: CALL    HOME.D          ;HOME SELECTED DRIVE.
1200    C0                  RNZ                     ;EXIT SEEK, HOME BAD.
1201    C3 1146             JMP     ..DTAS          ;NOW SEEK TRACK.

                        ;*****************************************************
```

```
                        ;*********************************************************
                        ; RD.SEC IS THE SUBROUTINE THAT INTERACTS WITH THE     *
                        ; 179X-02 DURING READ SECTOR OPERATIONS. THIS SECTION *
                        ; INITIATES THE DISK TRANSFER, SERVICES THE CONTROLLER*
                        ; CHIP DURING DATA TRANSFER, AND TERMINATES OPERATION *
                        ; WHEN FINISHED.  ERROR DETECTION IS IMPLEMENTED AND  *
                        ; RETRIES ARE EXRCUTED IF DATA ERRORS ARE DETECTED.   *
                        ;*********************************************************

                        ;*******( INITIALIZE READ OPERATION )******************

1204    AF              RD.SEC: XRA     A               ;ZERO A REGISTER.
1205    32 1330                 STA     ERR.CT          ;ZERO ERROR COUNT.
1208    3A 1373                 LDA     CB.SEC          ;LOAD SECTOR NMBR.
120B    A9                      XRA     C               ;INVERT (1791-01).
120C    D306                    OUT     WD.SEC          ;SET SECTOR REGISTER.
120E    FD21 1223       ..RTRY: LXI     Y,..NMI         ;LOAD NMI VECTOR.
1212    2A 132E                 LHLD    BUF.ST          ;BUFFER START.
1215    3E88                    MVI     A,DC.RDS        ;READ SECTOR COMMAND.
1217    A9                      XRA     C               ;INVERT (1791-01).
1218    D304                    OUT     WD.CMD          ;ISSUE READ COMMAND

                        ;*******( DATA TRANSFER LOOP )***************************

121A    DB80            ..REPT: IN      XP.DSH          ;HOLD FOR DATA
121C    DB07                    IN      WD.DTA          ;INPUT DATA.
121E    A9                      XRA     C               ;INVERT (1791-01).
121F    77                      MOV     M,A             ;PUT INTO BUFFER
1220    23                      INX     H               ;BUMP BUFF POINTER
1221    18F7                    JMPR    ..REPT          ;GO FOR ANOTHER

                        ;*******( CHECK STATUS )*********************************

1223    E69D            ..NMI:  ANI     DM.RER          ;TEST FOR ERRORS.
1225    32 1377                 STA     CB.STS          ;SAVE READ STATUS.
1228    C8                      RZ                      ;RETURN COMPLETE.
1229    CD 1279                 CALL    CHK.RT          ;CHECK ABOUT RETRYS.
122C    28E0                    JRZ     ..RTRY          ;PERFORM RETRY.
122E    C9                      RET                     ;ERROR RETURN.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; WR.SEC  SUBROUTINE  INTERACTS  WITH  THE  FD179X-02 *
                        ; DURING WRITE SECTOR OPERATIONS. THIS SECTION        *
                        ; INITIATES THE DISK TRANSFER, SERVICES THE CONTROLLER*
                        ; CHIP, AND TERMINATES THE OPERATION. ERROR DETECTION *
                        ; IS IMPLEMENTED.                                     *
                        ;*********************************************************

                        ;*******( INITIALIZE WRITE OPERATION )*****************

122F    AF              WR.SEC: XRA     A               ;ZERO REGISTER.
1230    32 1330                 STA     ERR.CT          ;SET ERROR COUNTER.
1233    3A 1373                 LDA     CB.SEC          ;LOAD SECTOR NMBR.
1236    A9                      XRA     C               ;INVERT (1791-01).
1237    D306                    OUT     WD.SEC          ;SET SECTOR REGISTER.
1239    FD21 124E       ..RTRY: LXI     Y,..NMI         ;SET NMI RETURN.
123D    2A 132E                 LHLD    BUF.ST          ;BUFFER START.
1240    3EA8                    MVI     A,DC.WRS        ;LOAD WRITE SECTOR CMD.
1242    A9                      XRA     C               ;INVERT (1791-01).
1243    D304                    OUT     WD.CMD          ;ISSUE COMMAND.

                        ;*******( DATA TRANSFER LOOP )***************************

1245    DB80            ..REPT: IN      XP.DSH          ;HOLD FOR DATA REQ.
1247    7E                      MOV     A,M             ;GET DATA BYTE.
1248    A9                      XRA     C               ;INVERT (1791-01).
1249    D307                    OUT     WD.DTA          ;OUTPUT DATA BYTE.
124B    23                      INX     H               ;INCREMENT BUFF POINTER
124C    18F7                    JMPR    ..REPT          ;REPEAT SEQUECE

                        ;*******( CHECK STATUS )*********************************

124E    E6FD            ..NMI:  ANI     DM.WER          ;TEST FOR WRITE ERRORS.
1250    32 1377                 STA     CB.STS          ;STORE WRITE STATUS.
1253    C8                      RZ                      ;RETURN COMPLETE.
1254    CD 1279                 CALL    CHK.RT          ;CHECK ABOUT RETRYS.
1257    28E0                    JRZ     ..RTRY          ;PERFORM RETRY.
1259    C9                      RET                     ;ERROR RETURN.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; WR.TRK IS THE SUBROUTINE WHICH INITIATES A FORMAT    *
                        ; TRACK COMMAND (WRITE-TRACK 179X-02 TYPE 3).  THE    *
                        ; FORMATTING BYTE STREAM IS PROVIDED BY A PROGRAM     *
                        ; WHICH MUST BE PRESENT IN THE FORMAT BUFFER.         *
                        ;*********************************************************

                        ;*******( INITIALIZE WRITE TRACK )**********************

125A    FD21 1266   WR.TRK: LXI     Y,..NMI         ;LOAD NMI VECTOR.
125E    3EF0                MVI     A,DC.WRT        ;WRITE TRACK CMND.
1260    A9                  XRA     C               ;INVERT (1791-01).
1261    D304                OUT     WD.CMD          ;ISSUE COMMAND.
1263    C3 1708             JMP     FMT.PS          ;FORMAT PROG START.

                        ;*******( CHECK COMPLETION STATUS )********************

1266    E6E4        ..NMI:  ANI     DM.FER          ;TEST FOR ERRORS.
1268    47                  MOV     B,A             ;HOLD THIS STATUS.
1269    DB00                IN      BL.STS          ;INPUT BOARD STATUS.
126B    E640                ANI     BS.TSD          ;TEST TWO SIDED BIT.
126D    78                  MOV     A,B             ;RESTORE STATUS TO A.
126E    2002                JRNZ    ..EXIT          ;NOT ZERO IS ONE SIDED.
1270    F601                ORI     CS.TSD          ;OR IN TWO SIDED FLAG.
1272    32 1377     ..EXIT: STA     CB.STS          ;STORE FORMAT STATUS.
1275    22 137A             SHLD    CW.LNG          ;DISPLAY TRAIL BYTES.
1278    C9                  RET                     ;RETURN TO USER.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; CHK.RT IS THE SUBROUTINE USED BY RD.SEC AND           *
                        ; WR.SEC TO COUNT RETRY OPERATIONS AND PERFORM A       *
                        ; RE-SEEK OPERATION WHEN NEEDED.                       *
                        ;*********************************************************

                        ;*******( CHECK IF RECOVERABLE )************************

1279    E680            CHK.RT: ANI     DM.DNR          ;TEST NOT READY BIT.
127B    2028                    JRNZ    ..EXIT          ;CAN NOT RECOVER.
127D    3A 137C                 LDA     CB.MOD          ;GET COMMAND MODE.
1280    E680                    ANI     CM.NRT          ;NO RETRYS CHECK.
1282    2021                    JRNZ    ..EXIT          ;SHOULD NOT RECOVER.
1284    DB40                    IN      XP.MTX          ;MOTOR TIME EXTEND.

                        ;*******( RECORD RETRY )*******************************

1286    3A 1330                 LDA     ERR.CT          ;GET ERROR COUNT.
1289    3C                      INR     A               ;INCREMENT.
128A    32 1330                 STA     ERR.CT          ;STORE NEW COUNT.
128D    FE05                    CPI     RTY.SK          ;SHOULD TRY SEEK?
128F    2008                    JRNZ    ..CKLS          ;IF NOT, CHECK LAST.

                        ;*******( REPOSITION R/W HEAD )************************

1291    CD 12A6                 CALL    HOME.D          ;HOME SELECTED DRIVE.
1294    200F                    JRNZ    ..EXIT          ;ERROR EXIT.
1296    CD 1131                 CALL    SEEK            ;SEEK DESIRED TRACK.

                        ;*******( HOLD READ GATE FOR 3/4 REVOLUTION )***********

1299    FE09            ..CKLS: CPI     RTY.LS          ;WAS THIS THE LAST.
129B    2807                    JRZ     ..STNZ          ;ERROR LAST RETRY.
129D    ED5B 1336               LDED    TM.PLD          ;PHASE LOCK DELAY.
12A1    CF                      WAIT                    ;PROGRAMMABLE DELAY.
12A2    AF                      XRA     A               ;CLEAR FOR RETRY.
12A3    C9                      RET                     ;TRY AGAIN EXIT.

                        ;*******( ERROR EXIT )*********************************

12A4    3C              ..STNZ: INR     A               ;SET NOT ZERO.
12A5    C9              ..EXIT: RET                     ;ERROR EXIT.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; HOME.D IS THE SUBROUTINE THAT STEPS THE DISK DRIVE    *
                        ; R/W HEAD OUTWARD UNTIL THE TRACK 0 FLAG BECOMES      *
                        ; ACTIVE OR 255 STEPS HAVE BEEN ISSUED.               *
                        ;*********************************************************

                        ;*******( RESTORE R/W HEAD )*****************************

12A6    3A 1331     HOME.D: LDA     SV.DRV          ;LOAD DRV NMBR ENABLED.
12A9    D300                OUT     BL.CTL          ;ISSUE CONTROLS.
12AB    32 1333             STA     SV.CTL          ;AND SAVE THESE.
12AE    2EFF                MVI     L,255           ;SET STEP COUNTER.
12B0    CD 1024     ..STEP: CALL    EX.STS          ;CHECK DISK STATUS.
12B3    E604                ANI     DM.TK0          ;INSPECT TRACK 0 FLG.
12B5    200C                JRNZ    ..EXIT          ;IF SET, GO ..EXIT.
12B7    2D                  DCR     L               ;DECREMENT STEP COUNT.
12B8    2814                JRZ     ..EROR          ;ERROR IF 255 STEPS.
12BA    DB08                IN      XP.STP          ;ISSUE STEP PULSE.
12BC    ED5B 1012           LDED    TM.STP          ;LOAD STEP DELAY.
12C0    CF                  WAIT                    ;PROGRAMMABLE DELAY.
12C1    18ED                JMPR    ..STEP          ;TRY STEPPING AGAIN.

                        ;*******( DRIVE IS RESTORED )***************************

12C3    ED5B 1014   ..EXIT: LDED    TM.ALS          ;TIME AFTER LAST STEP.
12C7    CF                  WAIT                    ;PROGRAMMABLE DELAY.
12C8    79                  MOV     A,C             ;GET WD TRK 0 VALUE.
12C9    D305                OUT     WD.TRK          ;ZERO TRACK REGISTER.
12CB    AF                  XRA     A               ;ZERO A REG, SET FLAG.
12CC    DD7701              MOV     DV.TRK(X),A     ;SET TRACK VALUE.
12CF    C9                  RET                     ;RETURN TO CALLER.

                        ;*******( TRACK 0 NOT FOUND )***************************

12D0    3E02        ..EROR: MVI     A,CS.HME        ;LOAD HOME ERROR FLAG.
12D2    32 1377             STA     CB.STS          ;STORE ERROR STATUS.
12D5    A7                  ANA     A               ;SET RETURN FLAGS.
12D6    C9                  RET                     ;RETURN TO CALLER.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; LOG.ON IS THE SUBROUTINE THAT READS THE IDENTITY     *
                        ; SECTOR FROM THE DISKETTE AND MAKES THE NEEDED       *
                        ; ENTRYS INTO THE DRIVE TABLE.  THE SECTOR DATA IS    *
                        ; ALSO LEFT IN THE SECTOR BUFFER FOR BIOS TO FINISH   *
                        ; THE LOG-ON OPERATION.                               *
                        ;*********************************************************

                        ;*******( CHECK JADE IDENTITY )*************************

12D7    11 133A     LOG.ON: LXI     D,JADEID        ;ID ADDRESS LOADED.
12DA    21 1380             LXI     H,ID.LBL        ;SECTOR ID ADDRESS.
12DD    0608                MVI     B,ID.SZE        ;ID LABEL SIZE.
12DF    1A          ..CKJI: LDAX    D               ;GET CHARACTER.
12E0    BE                  CMP     M               ;CHECK AGAINST DISK.
12E1    200B                JRNZ    ..3740          ;IF DIFFERENT: 3740.
12E3    13                  INX     D               ;CHECK NEXT.
12E4    23                  INX     H               ;CHECK NEXT.
12E5    10F8                DJNZ    ..CKJI          ;REPEAT OPERATION.

                        ;*******( LOG-ON JADE FORMAT )*************************

12E7    3A 13B1             LDA     ID.FLG          ;SIDE AND DENSITIES.
12EA    DD7702              MOV     DV.FLG(X),A     ;STORE IN DRIVE TBL.
12ED    C9                  RET                     ;RETURN TO CALLER.

                        ;*******( ASSUME 3740 FORMAT )*************************

12EE    3E00        ..3740: MVI     A,ID.FLD        ;SIDE AND DENSITIES.
12F0    DD7702              MOV     DV.FLG(X),A     ;STORE IN DRIVE TBL.
12F3    C9                  RET                     ;RETURN TO CALLER.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING ROUTINE SENDS ONE 8 BIT CHARACTER OUT *
                        ; THE EIA LEVEL TRANSMISSION BIT.  SET FOR BAUD RATE. *
                        ;*********************************************************

                        ;*******( SET UP FOR TRANSMISSION )*********************

12F4    DB00    LST.OT: IN      BL.STS          ;GET BOARD STATUS.
12F6    E610            ANI     BS.EIA          ;TEST LIST READY BIT.
12F8    CA 12F4         JZ      LST.OT          ;WAIT READY (JZ/JNZ).
12FB    3A 1375         LDA     CB.CHR          ;GET LIST CHARACTER.
12FE    2F              CMA                     ;COMPLEMENT ACUMULATOR.
12FF    5F              MOV     E,A             ;CHARACTER TO E REG.
1300    3A 1333         LDA     SV.CTL          ;LAST CONTROLS USED.

                        ;*******( SEND THE START BIT )*************************

1303    37              STC                     ;SET CARRY BIT.
1304    CD 131A         CALL    BIT.OT          ;OUTPUT START BIT.
1307    00              NOP                     ;EQUALIZE TIMING.
1308    00              NOP                     ;EQUALIZE TIMING.
1309    1608            MVI     D,8             ;NUMBER OF DATA BITS.

                        ;*******( SEND EACH DATA BIT )*****( 39 CYCLE LOOP )***

130B    CB0B    ..DATA: RRCR    E               ;ROTATE E REG RIGHT.
130D    CD 131A         CALL    BIT.OT          ;SEND ONE DATA BIT.
1310    15              DCR     D               ;ONE LESS BIT TO DO.
1311    C2 130B         JNZ     ..DATA          ;REPEAT IF MORE BITS.

                        ;*******( SEND STOP BIT )******************************

1314    00              NOP                     ;EQUALIZE TIMING.
1315    A7              ANA     A               ;CLEAR CARRY FLAG.
1316    CD 131A         CALL    BIT.OT          ;SEND STOP BIT.
1319    C9              RET                     ;RETURN TO CALLER.

                        ;*******( SET EIA BIT AND OUTPUT )****( 39 CYCLES )****

131A    DA 1322 BIT.OT: JC      ..ONE           ;IF CARRY, SET TO ONE.
131D    CB9F            RES     3,A             ;ZERO EIA IN ACUM REG.
131F    C3 1327         JMP     ..OUT           ;GO TO OUTPUT PORT.
1322    CBDF    ..ONE:  SET     3,A             ;SET EIA IN ACUM.
1324    C3 1327         JMP     ..OUT           ;EQUALIZE TIMING.
1327    D300    ..OUT:  OUT     BL.CTL          ;SEND ACUM TO PORT.

                        ;*******( SET DELAY FOR BAUDRATE )*********************

1329    0619            MVI     B,BAUD.C        ;LOAD TIMING CSNT.
132B    10FE            DJNZ    .               ;DELAY FOR BIT.
132D    C9              RET                     ;RETURN TO LST CALL.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; PROGRAM STORAGE LOACTIONS                              *
                        ;*********************************************************

132E    1380            BUF.ST: .WORD   BUF.BG  ;BUFFER STARTING ADDRESS.
1330    00              ERR.CT: .BYTE   0       ;RETRY ERROR COUNTER.

1331    00              SV.DRV: .BYTE   0       ;BL.CTL DRIVE BITS.
1332    00              SV.DAS: .BYTE   0       ;BL.CTL DRIVE AND SIDE BITS.
1333    00              SV.CTL: .BYTE   0       ;BL.CTL LAST ISSUED.
1334    00              SV.STS: .BYTE   0       ;FD179X-02 STATUS VALUE.

1335    00              PH.TRK: .BYTE   0       ;PHYSICAL TRACK NUMBER.

                        ;*********************************************************
                        ; TIMING VALUES - 0.1 MS INCREMENTS                      *
                        ;*********************************************************

1336    04B0            TM.PLD: .WORD   1200    ;PHASE LOCK RECOVERY.
1338    000A            TM.SAW: .WORD   10      ;STEP AFTER WRITING.
0018                    TM.SDD  ==      24      ;SIDE SELECT DELAY.

                        ;*********************************************************
                        ; DISKETTE IDENTITY LABEL                                *
                        ;*********************************************************

133A    4A6164652044 JADEID: .ASCII  "JADE DD "     ;DISKETTE ID LABEL.
0008                    ID.SZE  ==      (. - JADEID)    ;ID LABEL SIZE.

1380                    ID.LBL  ==  BUF.BG+0000H        ;ID SECTOR LABEL.
13A0                    ID.BLK  ==  ID.LBL+0020H        ;ID BLOCK AREA.
13B1                    ID.FLG  ==  ID.BLK+0011H        ;DISKETTE FLAGS.
0000                    ID.FLD  ==  00000000B           ;3740 FLAGS.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; DRIVE TABLE AREA DEFINED                               :
                        ;*********************************************************

                        ;*******( DRIVE TABLE ENTRIES )*************************

0000                    DV.NBR  ==      0       ;CURRENT DRIVE NUMBER.
0001                    DV.TRK  ==      1       ;CURRENT TRACK NUMBER.
0002                    DV.FLG  ==      2       ;SIDE AND DENSITY FLAGS
0003                    DV.CTL  ==      3       ;LAST CONTROLS USED.

                        ;*******( DRIVE TABLE AREA )****************************

1342                    DV.TBL  ==      .       ;DRIVE TABLE BEGGINING ADDRESS.

1342    00FF02C4        DT.DE0: .BYTE   0,255,DF.DFL,0C4H       ;DRIVE 0.
1346    01FF02C5        DT.DE1: .BYTE   1,255,DF.DFL,0C5H       ;DRIVE 1.
134A    02FF02C6                .BYTE   2,255,DF.DFL,0C6H       ;DRIVE 2.
134E    03FF02C7                .BYTE   3,255,DF.DFL,0C7H       ;DRIVE 3.
1352    04FF0000        DT.DED: .BYTE   4,255,0,0               ;DUMMY.

0004                    DV.DES  ==      DT.DE1-DT.DE0   ;EACH DRIVE ENTRY SIZE.

                        ;*******( FLAG BIT DEFINITIONS )***********************

0002                    DF.T1D  ==  00000010B   ;TRACK 1 DENSITY (1 = DOUBLE).
0004                    DF.DTD  ==  00000100B   ;DATA TRACKS DENSITY (1 = DD).
0008                    DF.TSD  ==  00001000B   ;TWO SIDED ( 1 = TWO SIDES).
0002                    DF.DFL  ==  DF.T1D      ;DEFAULT FLAGS.

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THE FOLLOWING AREA IS DEFINED AS THE COMMAND BLOCK.  *
                        ; THIS AREA IS RESERVED FOR SPECIFICATION BY THE HOST  *
                        ; SYSTEM FOR ALL DISK OPERATIONS.   CONTROLLER STATUS  *
                        ; AT COMPLETION OF OPERATION IS PRESENT IN THIS AREA.  *
                        ;*********************************************************

1370                                .LOC    CMD.BK  ;COMMAND BLOCK.

1370    00          CB.CMD: .BYTE   0       ;CONTROL COMMAND.
1371    00          CB.DRV: .BYTE   0       ;DRIVE NUMBER.
1372    00          CB.TRK: .BYTE   0       ;LOGICAL TRACK NUMBER.
1373    00          CB.SEC: .BYTE   0       ;SECTOR NUMBER.
1374    00          CB.FFG: .BYTE   0       ;FORMAT FLAGS.
1375    00          CB.CHR: .BYTE   0       ;EIA CHARACTER.
1376    00          CB.MOD: .BYTE   0       ;MODE SELECTS.
1377    00          CB.STS: .BYTE   0       ;CONTROLLER STATUS.

1378    0000        CW.LAD: .WORD   0       ;LOAD ADDRESS.
137A    0000        CW.LNG: .WORD   0       ;LOAD LENGTH

                        ;*******( MODE BIT DEFINITIONS )************************

0080                CM.NRT  ==  10000000B   ;NO RETRYS ( = 1 ).

                        ;*******( STATUS BIT DEFINITIONS )**********************

0080                CS.DNR  ==  10000000B   ;DRIVE NOT READY.
0040                CS.WRP  ==  01000000B   ;WRITE PROTECTED.
0020                CS.BT5  ==  00100000B   ;NOT ASSIGNED.
0010                CS.RNF  ==  00010000B   ;RECORD NOT FOUND.
0008                CS.CRC  ==  00001000B   ;CRC ERROR.
0004                CS.LDE  ==  00000100B   ;LOST DATA ERROR.
0002                CS.HME  ==  00000010B   ;DRIVE HOME ERROR.
0001                CS.TSD  ==  00000001B   ;TWO SIDES FLAG (FORMAT).

                        ;*********************************************************
```

```
                        ;*********************************************************
                        ; THIS SECTION RESIDES IN THE DCM SECTOR BUFFER. THIS *
                        ; SECTION MOVES DCM FROM BANK 1 DOWN TO BANK 0.   THE *
                        ; C REGISTER IS SET FOR 1791-01 OR 1793-01.  THE LAST *
                        ; OPERATION IS TO  READ THE  BIOS  LOADER  SECTOR  TO *
                        ; OVERLAY THIS INITIALIZATION SEQUENCE.   BIOS LOADER *
                        ; THEN READ BIOS INTO BANK 1 AND HALTS.               *
                        ;*********************************************************

                        ;*******( EXECUTES IN BANK 1 )*************************

1380                            .LOC    BUF.BG          ;RESIDES IN BUFFER.
1380    01 0400         INIT.B: LXI     B,BANK.L        ;SET BANK LENGTH.
1383    11 1000                 LXI     D,BANK.0        ;SET DESTINATION.
1386    21 1400                 LXI     H,BANK.1        ;SET SOURCE ADDR.
1389    EDB0                    LDIR                    ;MOVE BLOCK.
138B    C3 138E                 JMP     ..DOWN          ;JUMP TO NEW IMAGE.

                        ;*******( NOW IN BANK 0, SET INT MODE )****************

138E    31 1370         ..DOWN: LXI     SP,TP.STK       ;SET STACK PNTR.
1391    ED56                    IM1                     ;INTERRUPT MODE 1.

                        ;*******( SET 1791-01/1793-01 )***********************

1393    0E00                    MVI     C,0             ;LOAD C REG ZERO.
1395    DB00                    IN      BL.STS          ;BOARD STATUS.
1397    E601                    ANI     BS.US0          ;TEST USER SW #1.
1399    2002                    JRNZ    LD.BLT          ;SW OPEN - 1793.
139B    0EFF                    MVI     C,0FFH          ;SW CLOSED - 1791.

                        ;*******( OVERLAY WITH BIOS LOADER TRANSIENT )*********

139D    DD21 1352       LD.BLT: LXI     X,DT.DED        ;INIT DRIVE TBL.
13A1    3E02                    MVI     A,2             ;BIOS LOADER SECTOR.
13A3    32 1373                 STA     CB.SEC          ;SET SECTOR VALUE.
13A6    DB40                    IN      XP.MTX          ;MOTOR TIME EXTEND.
13A8    21 1380                 LXI     H,BUF.BG        ;SET RETURN ADDR.
13AB    E5                      PUSH    H               ;PUSH INTO STACK.
13AC    C3 1204                 JMP     RD.SEC          ;GET BIOS LOADER.

                        ;*********************************************************
                                .END
```

+++++ SYMBOL TABLE +++++

BANK.0 1000	BANK.1 1400	BANK.L 0400	BASE 1000
BAUD.C 0019	BC.DAS 0020	BC.DDE 0010	BC.DDS 0010
BC.DSA 0001	BC.DSB 0002	BC.DSE 0004	BC.DSN 0003
BC.EIA 0008	BC.INW 0020	BC.PCA 0040	BC.PCB 0080
BC.PCH 0040	BC.PCL 00C0	BC.PCM 0080	BC.SD1 0020
BC.SDS 0000	BIT.OT 131A	BL.CTL 0000	BL.STS 0000
BS.DCN 0080	BS.EIA 0010	BS.INT 0008	BS.MOF 0020
BS.TSD 0040	BS.TST 0004	BS.US0 0001	BS.US1 0002
BUF.BG 1380	BUF.ST 132E	CB.CHR 1375	CB.CMD 1370
CB.DRV 1371	CB.FFG 1374	CB.MOD 1376	CB.SEC 1373
CB.STS 1377	CB.TRK 1372	CHK.RT 1279	CMD.BK 1370
CM.DTA 1053	CM.MSK 0007	CM.NRT 0080	CS.BT5 0020
CS.CRC 0008	CS.DNR 0080	CS.HME 0002	CS.LDE 0004
CS.RNF 0010	CS.TSD 0001	CS.WRP 0040	CW.LAD 1378
CW.LNG 137A	DC.HDL 0018	DC.HDU 0010	DC.IFI 00D8
DC.RDA 00C0	DC.RDS 0088	DC.STS 00D0	DC.WRS 00A8
DC.WRT 00F0	DF.DFL 0002	DF.DTD 0004	DF.T1D 0002
DF.TSD 0008	DM.DNR 0080	DM.FER 00E4	DM.HDL 0020
DM.LDE 0004	DM.RER 009D	DM.TK0 0004	DM.WER 00FD
DT.DE0 1342	DT.DE1 1346	DT.DED 1352	DV.CTL 0003
DV.DES 0004	DV.FLG 0002	DV.NBR 0000	DV.TBL 1342
DV.TRK 0001	ERR.CT 1330	EX.HCF 1018	EX.STS 1024
FETCH 103F	FMT.BG 1700	FMT.PS 1708	HOME.D 12A6
HR.INT 1038	HR.VEC 1006	ID.BLK 13A0	ID.FLD 0000
ID.FLG 13B1	ID.LBL 1380	ID.SZE 0008	INIT.B 1380
IO.BLK 1370	JADEID 133A	LD.BLT 139D	LOG.ON 12D7
LST.OT 12F4	NM.INT 1066	PH.TRK 1335	RD.SEC 1204
RST.0 1000	RST.1 1008	RST.2 1010	RST.3 1018
RST.4 1020	RST.5 1028	RST.6 1030	RST.7 1038
RTY.LS 0009	RTY.SK 0005	SEEK 1131	SELECT 10EF
SV.CTL 1333	SV.DAS 1332	SV.DRV 1331	SV.STS 1334
TICK.E 1074	TICK.R 1070	TMR.FC 0019	TMR.NC 001C
TM.ALS 1014	TM.HLD 1010	TM.MTO 1016	TM.PLD 1336
TM.SAW 1338	TM.SDD 0018	TM.STP 1012	TP.STK 1370
TRK.IB 0034	TRK.OB 001A	WD.CMD 0004	WD.DTA 0007
WD.INT 1066	WD.SEC 0006	WD.STS 0004	WD.TRK 0005
WR.SEC 122F	WR.TRK 125A	XP.DSH 0080	XP.IRR 0020
XP.MTO 0010	XP.MTX 0040	XP.STP 0008	X.CUTE 1041
$.ADDR 10C7	$.FORM 1098	$.IDLE 10E4	$.LGON 10AC
$.LIST 10CF	$.LSTT 10D5	$.READ 107C	$.WRIT 108A

```
F1000,13FF,0
-IDCM2.HEX
-R
NEXT  PC
13AF 0000
-D1000,12FF
1000 C3 00 00 C3 80 17 41 10 06 19 10 FE C3 74 10 00 ......A......T..
1010 5E 01 50 00 50 00 01 00 FD E1 DB 05 D3 07 78 A9 ^.P.P.........X.
1020 D3 04 18 FE 3E D0 A9 D3 04 E3 E3 E3 E3 DB 04 A9 ....>...........
1030 C9 00 00 00 00 00 00 00 DB 20 D1 2A 06 10 E9 FB ......... .*....
1040 76 3A 70 13 E6 07 87 16 00 5F 21 53 10 19 5E 23 V:P......_!S..^#
1050 56 EB E9 AC 10 7C 10 8A 10 98 10 C7 10 CF 10 D5 V....\..........
1060 10 E4 10 00 00 00 DB 04 A9 32 34 13 FD E3 ED 45 .........24....E
1070 06 1C 10 FE 1B 7A B3 00 00 20 F5 C9 CD EF 10 CD .....Z... ......
1080 31 11 20 03 CD 04 12 C3 3F 10 CD EF 10 CD 31 11 1. .....?.....1.
1090 20 03 CD 2F 12 C3 3F 10 CD EF 10 3A 73 13 DD 77  ../..?....:S..W
10A0 02 CD 31 11 20 03 CD 5A 12 C3 3F 10 CD EF 10 AF ..1. ..Z..?.....
10B0 32 72 13 3C 32 73 13 CD 31 11 20 08 CD 04 12 20 2R.<2S..1. ....
10C0 03 CD D7 12 C3 3F 10 3E FF 32 77 13 C3 3F 10 CD .....?.>.2W..?..
10D0 F4 12 C3 3F 10 DB 00 E6 10 CA DE 10 3E FF 32 77 ...?........>.2W
10E0 13 C3 3F 10 DB 00 E6 08 28 FA DB 20 C3 3F 10 DB ..?.....(.. .?..
10F0 00 E6 20 DB 40 28 05 ED 5B 16 10 CF 3A 71 13 E6 .. .@(..[...:Q..
1100 03 DD BE 00 C8 DD 21 42 13 11 04 00 3D FA 14 11 ......!B....=...
1110 DD 19 18 F8 06 10 CD 18 10 3A 33 13 E6 FB D3 00 .........:3.....
1120 E6 FC DD B6 00 D3 00 F6 04 D3 00 E6 07 32 31 13 .............21.
1130 C9 CD 24 10 E6 A0 FA 74 11 C2 46 11 06 18 CD 18 ..$....T..F.....
1140 10 ED 5B 10 10 CF DB 00 E6 40 3A 72 13 6F C2 52 ..[.......@:R.O.R
1150 11 1F 32 35 13 67 3A 31 13 30 02 F6 20 32 32 13 ..25.G:1.0.. 22.
1160 57 7C DD 96 01 20 21 DB 00 E6 40 CA 7B 11 DD 7E W\... !...@.[..^
1170 03 C3 EA 11 3E 80 32 77 13 A7 C9 7C A7 28 3B DD ....>.2W...\.(;.
1180 7E 03 E6 DF B2 C3 EA 11 F5 ED 5B 38 13 CF F1 38 ^.........[8...8
1190 0A 6F 3A 31 13 F6 20 D3 00 18 0B ED 44 FA FD 11 .O:1.. ......D...
11A0 6F 3A 31 13 D3 00 DB 08 ED 5B 12 10 CF 2D 20 F6 O:1.......[...- .
11B0 3A 32 13 D3 00 ED 5B 14 10 CF 3A 72 13 FE 01 38 :2....[...:R...8
11C0 20 3E 04 C2 C8 11 3E 02 DD A6 02 CA E1 11 3A 35  >....>........:5
11D0 13 FE 1A 06 D0 38 0C FE 34 06 90 38 06 06 50 18 .....8..4..8..P.
11E0 02 06 C0 3A 32 13 B0 DD 77 03 D3 00 32 33 13 3A ...:2...W...23.:
11F0 35 13 DD 77 01 3A 72 13 A9 D3 05 AF C9 CD A6 12 5..W.:R.........
1200 C0 C3 46 11 AF 32 30 13 3A 73 13 A9 D3 06 FD 21 ..F..20.:S.....!
1210 23 12 2A 2E 13 3E 88 A9 D3 04 DB 80 DB 07 A9 77 #.*..>.........W
1220 23 18 F7 E6 9D 32 77 13 C8 CD 79 12 28 E0 C9 AF #....2W...Y.(...
1230 32 30 13 3A 73 13 A9 D3 06 FD 21 4E 12 2A 2E 13 20.:S.....!N.*..
1240 3E A8 A9 D3 04 DB 80 7E A9 D3 07 23 18 F7 E6 FD >.......^...#....
1250 32 77 13 C8 CD 79 12 28 E0 C9 FD 21 66 12 3E F0 2W...Y.(...!F.>.
1260 A9 D3 04 C3 08 17 E6 E4 47 DB 00 E6 40 78 20 02 ........G...@X .
1270 F6 01 32 77 13 22 7A 13 C9 E6 80 20 28 3A 76 13 ..2W."Z.... (:V.
1280 E6 80 20 21 DB 40 3A 30 13 3C 32 30 13 FE 05 20 .. !.@:0.<20...
1290 08 CD A6 12 20 0F CD 31 11 FE 09 28 07 ED 5B 36 .... ..1...(..[6
12A0 13 CF AF C9 3C C9 3A 31 13 D3 00 32 33 13 2E FF ....<.:1...23...
12B0 CD 24 10 E6 04 20 0C 2D 28 16 DB 08 ED 5B 12 10 .$... .-(....[..
12C0 CF 18 ED ED 5B 14 10 CF 79 D3 05 AF DD 77 01 C9 ....[...Y....W..
12D0 3E 02 32 77 13 A7 C9 11 3A 13 21 80 13 06 08 1A >.2W....:.!.....
12E0 BE 20 0B 13 23 10 F8 3A B1 13 DD 77 02 C9 3E 00 . ..#..:...W..>.
12F0 DD 77 02 C9 DB 00 E6 10 CA F4 12 3A 75 13 2F 5F .W.........:U./_
-
```

```
D1300,13FF
1300 3A 33 13 37 CD 1A 13 00 00 16 08 CB 0B CD 1A 13 :3.7.............
1310 15 C2 0B 13 00 A7 CD 1A 13 C9 DA 22 13 CB 9F C3 ...........".....
1320 27 13 CB DF C3 27 13 D3 00 06 19 10 FE C9 80 13 '....'..........
1330 00 00 00 00 00 00 B0 04 0A 00 4A 61 64 65 20 44 ..........JADE D
1340 44 20 00 FF 02 C4 01 FF 02 C5 02 FF 02 C6 03 FF D ...............
1350 02 C7 04 FF 00 00 00 00 00 00 00 00 00 00 00 00 ................
1360 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
1370 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
1380 01 00 04 11 00 10 21 00 14 ED B0 C3 8E 13 31 70 ......!.......1P
1390 13 ED 56 0E 00 DB 00 E6 01 20 02 0E FF DD 21 52 ..V...... ....!R
13A0 13 3E 02 32 73 13 DB 40 21 80 13 E5 C3 04 12 00 .>.2S..@!.......
13B0 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
13C0 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
13D0 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
13E0 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
13F0 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
-
```